
― 原著論文 ―

ベクトル型計算機によるPIC（Particle-In-Cell）法速度モーメント計算の計算速度比較

～作業配列法・RETRY法・LISTVEC指示行法～

杉山　徹1*

粒子シミュレーション手法の1つであるPIC（Particle-In-Cell）法モデルにおいて，粒子の速度モーメントをベクトル型

計算機上で計算するためには，いくつかの工夫が必要である．その中で，広く使われている3つの手法（作業配列法・

RETRY法・LISTVEC指示行法）による計算速度を比較した結果を報告する．計測には空間2次元システムを用いた．作

業配列法・RETRY法の計算速度は同程度であるが，LISTVEC指示行法はそれらより約1.5倍遅い．一方，使用する主メ

モリ量は，RETRY法・LISTVEC指示行法は同程度であるが，作業配列法は，それらより数倍多く，扱えるシステムサ

イズを小さくしてしまう．プログラミングの容易さでは，LISTVEC指示行法は非常に容易であり，次に作業配列法であ

る．RETRY法はプログラム作成までの作業量が他の2つに比べて多い．結果，本報告では，プログラミングは容易では

ないが，計算効率を全体的に判断し，RETRY法の使用を推薦する．

ベクトル型計算機によるPICモーメント計算法の速度比較
Performance measurement of PIC moment calculation

― Original Paper ―

Performance measurement of WorkArray, LISTVEC compile directive, and RETRY algorithms

on SX Super-computer for the PIC (Particle In Cell) simulation model

Tooru Sugiyama1*

The calculation performance of three types of moment calculation algorithms is evaluated on SX super computer. Those are

WorkArray, LISTVEC compile directive, and RETRY algorithms. The used simulation model consists of two dimensional in space.

The calculation time of WorkArray and RETRY methods show almost same cost. That of LISTVEC methods is about 1.5 times larger

than those of others. The used main memory size of RETRY and LISTVEC methods is almost same cost. That of WorkArray is a few

times larger than those of others. The algorithm of WorkArray and LISTVEC are simple but that of RETRY is complicated. However,

in the view points of total merits, we recommend to use RETRY algorithm.

1．はじめに

PIC（Particle-in-Cell）シミュレーション手法は，プラズマ

運動論計算において重要なモデルの1つである（例えば,

Birdsall and Langdon, 1991; Fujimoto, 1992; Matsumoto and Omura,

1993）．この手法では，計算領域内において超粒子と呼ば

れる仮想粒子の軌道を追う．一方，密度・流量などの流体

量や電磁場は格子点上で解かれる．そのため，粒子の速度

モーメント値（電荷密度・電流密度など）は，粒子位置に

隣接する格子点と関連付けられなければならない．ここで

問題となるのが，計算領域内のランダムな位置に配置され

ている粒子データに対し，どのようにベクトル型計算機で

モーメント値を計算するかである．これまでに，作業配列

法（Birdsall and Langdon, 1991）・RETRY法（Imamura,

2005）・LISTVEC指示行法（Sugiyama et al., 2004）の3つの

方法が提案されており，ここでは，これらの方法の実行速

度を測定した結果を報告する．使用したベクトル型計算機

は，2009年から稼働している地球シミュレータである．用

いたコードは，空間2次元・速度3次元で構成されるシステ

ムを取り扱ったコードである．

2．粒子モーメントの計算法

粒子法では，位相空間内での粒子の分布関数を滑らか

に表現し，運動論を正しく議論するために，格子点当り

に百個程度の粒子を分布させる．よって，格子点上での

電磁場に関する計算量に比べて，粒子に関する計算量が

多い．このことから，計算効率の向上には，粒子計算部

を効率良く計算する必要がある．その中で，最も計算コ

ストの高い作業は，システム上にランダムに配置された

粒子と，格子点上で定義された場の量とを関連付ける計

算（モーメント計算）である．この計算は，格子点位置r

における電荷密度は，

電流密度は，

と定義される．ここで，r#, v#は粒子の位置と速度であり，

全粒子の総和計算を行う．S(r)は，粒子と格子を関係付け

る形状関数である．図1に形状関数の概念図を示す．（図

の説明では，簡単のため，1次精度1次元モデルで行う）．

超粒子は，点電荷としてではなく，粒子座標 Xpを中心と

する格子間隔（ΔX）と同じ長さの広がりを持った粒子と

Fig. 1.   Schematic illustration of the shape function in one-dimensional case. The particle width (dark gray length of a) which lies in the ith cell

contributes to the moment on the grid Xi.

図1. 形状関数の概念図．格子点座標をXi，粒子位置をXpとする．粒子は，格子点幅ΔXと同じ長さの広がりを持ってシステム内に存在す
る．各格子点は，幅ΔXの領域を担当し，その中に入ってきた粒子の長さ分を，自身の格子点上でのモーメント値に寄与させる．図の例で
は，長さa分が格子点Xiへの寄与分となる．

N(r)＝ S(r－r#)
#
Σ 

F(r)＝ v#S(r－r#)
#
Σ 

して扱われる．また，各格子は，格子点位置（Xi = i ΔX）

を中心とする幅（ΔX）の領域を自身の領域とする．領域

内に属した長さ分を，粒子の格子点へのモーメント量の

寄与分とする．図の例では，格子Xi とXi+1への寄与分は，

それぞれa/ΔX，1- a/ΔXとなる．

プログラム例を図２に示す．この例に見られるように，

粒子と格子点の座標を格子間隔ΔXで規格化してからシミ

ュレーションを実行するため，格子点への寄与分を計算

する14～15行目に，ΔXは陽には表れない．

この例は，スカラ型計算機向けのコードである．その

ため，このままでは，メインループ（12～26行目）にお

いて，粒子位置がランダムなため，粒子のID番号NNに対

応する格子点位置IMの依存関係が不明となり，ベクトル

計算が出来ない．

以下，提案されている3つのベクトル化方法について述

べる．

2.1．作業配列法

電荷密度と電流密度を保存する配列のインデックスを，

もう1次元拡張し，その要素数でループ計算を行うことで

依存関係を解消する．別名「とまり木」法とも呼ばれる．

プログラム例を図3に示す．拡張した配列は，7行目に定

義されている．この方法では，拡張したインデックスの

サイズが最大ベクトルレジスタ長（以下ベクトル長，

ベクトル型計算機によるPICモーメント計算法の速度比較
Performance measurement of PIC moment calculation

!!    PROGRAM  SCALAR

!!

REAL:: PTL(1:NMAX,1:4)           ! Particle

REAL:: MOM(1:IMAX,1:4)           ! Moment

MOM = 0.0

!!

!!  MAIN LOOP

!!

DO NN = 1, NMAX

IM = FLOOR( PTL(NN,4) );    IP=1  +IM      !.. Grid in X

XM = PTL(NN,4)-REAL(IM);    XP=1.0-XM      !.. Shape

MOM(IM,1) = MOM(IM,1) +  XM * PTL(NN,1) !.. Flux in X

MOM(IP,1) = MOM(IP,1) +  XP * PTL(NN,1)

MOM(IM,2) = MOM(IM,2) +  XM * PTL(NN,2) !.. Flux in Y

MOM(IP,2) = MOM(IP,2) +  XP * PTL(NN,2)

MOM(IM,3) = MOM(IM,3) +  XM * PTL(NN,3) !.. Flux in Z

MOM(IP,3) = MOM(IP,3) +  XP * PTL(NN,3)

MOM(IM,4) = MOM(IM,4) +  XM !.. Density

MOM(IP,4) = MOM(IP,4) +  XP

ENDDO

Fig. 2.   Program for Scalar procedure. NMAX and IMAX mean the total number of particles and cells. PTL(particle ID num, 1~4) means particle

velocity with three components and position. The moment values on grid are stored in MOM(grid ID, 1~4) with three components of the flux and the

density.

図2. モーメント計算のプログラム例．スカラ型計算機で実行する場合のプログラミング方法である．各変数は，粒子総数がNMAX．格子
点総数がIMAX．粒子の速度3成分がPTL（粒子ID番号，1～3），粒子位置がPTL（粒子ID番号，4），モーメント値のFlux値がMOM（格子番
号，1～3），密度がMOM（格子番号，4），である．

Fig. 3.   Program for WORKARRAY algorithm. The extended array is defined as WRK(Grid ID, extended index, components).

図3. 作業配列法によるモーメント計算のプログラム例．拡張した作業配列は，WRK（格子番号，拡張index，成分）と定義されている．

!!

!!    PROGRAM  WORK VECTOR

!!

INTEGER,PARAMETER:: LVEC=256

REAL::  PTL(1:NMAX,1:4) 

REAL::  MOM(1:IMAX,1:4)

REAL::  WRK(1:IMAX,1:LVEC,1:4)

!!

!!   INITIALIZATION  LOOP

!!

MOM = 0.0

WRK = 0.0

!!

!!   MAIN  LOOP

!!

DO NP =  1, NMAX, LVEC

DO NN = NP, MIN( NP+LVEC, NMAX)

NV = NN - NP + 1

IM = FLOOR( PTL(NN,4) ); IP=1  +IM      !.. Grid in X

XM = PTL(NN,4)-REAL(IM); XP=1.0-XM      !.. Shape

WRK(IM,NV,1) = WRK(IM,NV,1) + XM * PTL(NN,1)

WRK(IP,NV,1) = WRK(IP,NV,1) + XP * PTL(NN,1)

WRK(IM,NV,2) = WRK(IM,NV,2) + XM * PTL(NN,2)

WRK(IP,NV,2) = WRK(IP,NV,2) + XP * PTL(NN,2)

WRK(IM,NV,3) = WRK(IM,NV,3) + XM * PTL(NN,3)

WRK(IP,NV,3) = WRK(IP,NV,3) + XP * PTL(NN,3)

WRK(IM,NV,4) = WRK(IM,NV,4) + XM

WRK(IP,NV,5) = WRK(IP,NV,4) + XP

ENDDO

ENDDO

!!

!!   REDUCTION  LOOP

!!

DO IX = 1, IMAX

DO IC = 1, 4

DO NV = 1, LVEC

MOM(IX,IC) =  MOM(IX,IC) + WRK(IX,NV,IC)

ENDDO

ENDDO

ENDDO

NEC-SXシリーズでは256）の場合（4行目：LVEC=256）に

計算速度が最速である．この時のサイズは，超粒子を格

子点当り256×4個分布させることと同値であり（×4は，

電荷密度と電流密度の4成分用），主メモリ使用量を圧迫

してしまう．その使用メモリ量を，粒子用メモリに使用

すれば，扱う粒子数が増え，さらに統計的に精密な運動

論計算が可能となる．さらに，粒子数に関する総和計算

前後に，初期化と総和計算（reduction）が必要である（13

行目と41～47行目）．この計算量は，格子点数が多い場合

には，無視できない計算コストを生む．後述にあるよう

に，格子点数が約107の時は，メインループの7％にも達す

る．

2.2．LISTVEC指示行法

作業配列の導入による主メモリ使用量の増加を回避す

る方法の一つが，NEC-SXシリーズ独特のコンパイル指示

行LISTVECを用いる方法である（内容の詳細は文献

Sugiyama et al., 2004を参照のこと）．概要は，図2における

メインループ内で，異なる粒子ID番号に対して，同じ格

子番号IMが割り当てられる時に（衝突発生と呼ぶ），結果

が不整合となるが，衝突が起こらない限り正しい結果を

得る．よって，衝突発生を監視し，衝突が発生した時の

みベクトル計算からスカラ計算に切り替えれば正しい結

果を得る．プログラム例を図4に示す．この指示行（22行

目）が記載された直後のループでは，自動的に衝突発生

を監視しながら和計算を行う．本指示行を用いると，衝

突発生作業配列としての主メモリの圧迫量が，格子点当

り2次元計算の場合は4×4個（１次元計算の場合は2×4個）

のみの粒子を分布させる量と同値であるため大幅に軽減

され，かつ図2にあるスカラ型プログラムからの変更は少

ない．（注：補正作業を行うため，ループ内で同じ名前の

配列を複数行記載できないため，6～9行目にあるように

配列名が増えている）．また，作業配列法に比べ，

reduction 作業による計算コストは格段に小さい．

2.3．Retry 法

この方法は，NEC-SXシリーズに限らず，一般的なベク

トル型計算機で使用できる方法である．ここでは，「ベク

トル型計算機では，有限長のベクトルレジスタ上のデー

タがメモリにストアされる時，書き込み順は常に，一定

の順序で行われる」というベクトル型計算機のハードウ

ェア特性を利用する（Imamura, 2005）．アルゴリズムの概

念図とプログラム例を図5と6に示す．

本アルゴリズムでは，全粒子をRVEC個の束にしてルー

プを考える．図5の例ではRVEC=12である．（注：RVECの

大きさは，ベクトル長とは独立であるが，ベクトル長以

上でなければ高速計算ができない）．まず，図6の24～28行

目では，粒子のID番号NNに対応する格子点位置IMを求め，

配列CIG（IM）へ粒子番号を記録する（モーメント値を書

き込むのではない）．メモリへの書き込み順が一定である

という特性から，この書き込みでは，格子点位置の衝突

が発生した場合は，自動的に，粒子ID番号の大きい方の

データが配列CIGに記録される（図5の上部の灰色網掛け

部分）．配列CIGは，格子点当り粒子1個を記録できれば十

分であるが，KVEC個を記録できるようインデックスを拡

張している．KVEC（=NKの値が取り得る範囲．図5の例

では3．）の値を大きくすれば，明らかに，その衝突頻度

は下がるが，作業配列法と同じ衝突解消方法であるため，

主メモリ使用量が増えてしまう．次に，30～53行目にお

いて，モーメント値を計算して，配列TRYmとTRYpに値

を代入していく．ここで，上記24～28行目と同じループ

を回し，衝突発生をチェックする．35行目のIF文において，

衝突が発生している場合は，配列CIGの値と粒子ID番号が

異なるため判別可能である．衝突が発生している場合は，

その粒子ID番号を配列CINへスタックする（49～50行目）．

以降は，このスタックされた粒子群に対して，同じ事

を繰り返し，衝突を起こしスタックされる粒子が無くな

るまでDO WHILE－ENDDOで反復処理（RETRY）を行う．

最後に，拡張したインデックスの要素に対する総和計

算（reduction）を行う．

ベクトル型計算機によるPICモーメント計算法の速度比較
Performance measurement of PIC moment calculation

Fig. 4.   Program for LISTVEC algorithm.

図4. LISTVEC法によるモーメント計算のプログラム例．メインループの直前にあたる22行目に，指示行が入れられている．

!!

!!    PROGRAM  LISTVEC

!!

REAL::  PTL(1:NMAX,1:4) 

REAL::  MOM(1:IMAX,1:4)

REAL::  WXM(1:IMAX), WXP(1:IMAX)

REAL::  WYM(1:IMAX), WYP(1:IMAX)

REAL::  WZM(1:IMAX), WZP(1:IMAX)

REAL::  WNM(1:IMAX), WNP(1:IMAX)

!!

!!   INITIALIZATION  LOOP

!!

WXM = 0.0;  WXP = 0.0 !..  Flux in X

WYM = 0.0;  WYP = 0.0 !..  Flux in Y

WZM = 0.0;  WZP = 0.0 !..  Flux in Z

WNM = 0.0;  WNP = 0.0 !..  Density

!!

!!   MAIN  LOOP

!!

!CDIR  LISTVEC

DO NN =  1, NMAX

IM = FLOOR( PTL(NN,4) );    IP=1  +IM     !.. Grid in X

XM = PTL(NN,4)-REAL(IM);    XP=1.0-XM     !.. Shape

WXM(IM) = WXM(IM) + XM * PTL(NN,1)

WXP(IP) = WXP(IP) + XP * PTL(NN,1)

WYM(IM) = WYM(IM) + XM * PTL(NN,2)

WYP(IP) = WYP(IP) + XP * PTL(NN,2)

WZM(IM) = WZM(IM) + XM * PTL(NN,3)

WZP(IP) = WZP(IP) + XP * PTL(NN,3)

WNM(IM) = WNM(IM) + XM

WNP(IP) = WNP(IP) + XP

ENDDO

!!

!!  REDUCTION  LOOP

!!

DO IX  = 1, IMAX

MOM(IX,1) = WXM(IX) + WXP(IX)

MOM(IX,2) = WYM(IX) + WYP(IX)

MOM(IX,3) = WZM(IX) + WZP(IX)

MOM(IX,4) = WNM(IX) + WNP(IX)

ENDDO

Fig. 5.   Conceptual illustration for RETRY algorithm.

図5. RETRY法の概念図．

Fig. 6.   Program for RETRY algorithm. The NODEP compile-directive is necessary for vectorized operation.

図6. RETRY法によるモーメント計算のプログラム例．途中NODEP指示行を入れなければ，コンパイラは依存関係があるとしてループが
ベクトル化されない．

!!

!!    PROGRAM  RETRY

!!

INTEGER,PARAMETER::   RVEC = 2048, KVEC=2

REAL::  PTL(1:NMAX,1:4) 

REAL::  MOM(1:IMAX,1:4)

REAL::  TRYm(1:KVEC,1:IMAX,1:4), TRYp(1:KVEC,1:IMAX,1:4)

INTEGER::  CIG(1:IMAX,1:KVEC)

INTEGER::  CIN(1:NMAX)

INTEGER::  ct, ca

!!

!!   INITIALIZATION  LOOP

!!

TRYm = 0.0;  TRYp = 0.0

MOM  = 0.0

!!

!!   MAIN  LOOP

!!

ct = 0

DO NP = 1, NMAX, RVEC

DO NN = NP, MIN( NP+RVEC-1, NMAX)

NK = MOD(NN,KVEC) + 1

IM = FLOOR( PTL(NN,4) )

CIG(IM,NK) = NN

ENDDO

!CDIR NODEP

DO NN = NP, MIN( NP+RVEC-1, NMAX)

NK = MOD(NN,KVEC) + 1

IM = FLOOR( PTL(NN,4) )

IF( CIG(IM,NK) == NN ) THEN

IP = IM + 1

XM = PTL(NN,4)-REAL(IM)

XP = 1.0 - XM

TRYm(NK,IM,1) = TRYm(NK,IM,1) + XM * PTL(NN,1)

TRYp(NK,IP,1) = TRYp(NK,IP,1) + XP * PTL(NN,1)

TRYm(NK,IM,2) = TRYm(NK,IM,2) + XM * PTL(NN,2)

TRYp(NK,IP,2) = TRYp(NK,IP,2) + XP * PTL(NN,2)

TRYm(NK,IM,3) = TRYm(NK,IM,3) + XM * PTL(NN,3)

TRYp(NK,IP,3) = TRYp(NK,IP,3) + XP * PTL(NN,3)

TRYm(NK,IM,4) = TRYm(NK,IM,4)

TRYp(NK,IP,4) = TRYp(NK,IP,4)

ELSE

ct = ct + 1

CIN(ct) = nn

ENDIF

ENDDO

ENDDO

!-----

ca = ct

DO WHILE( 0 < ca )

ct = 0

DO NP = 1, ca, RVEC

DO NV = NP, MIN( NP+RVEC-1, ca)

NN = CIN(NV)

NK = MOD(NN,KVEC) + 1

IM = FLOOR( PTL(NN,4) )

CIG(IM,NK) = NN

ENDDO

!CDIR NODEP

DO NV = NP, MIN( NP+RVEC-1, ca)

NN = CIN(NV)

NK = MOD(NN,KVEC) + 1

IM = FLOOR( PTL(NN,4) )

IF( CIG(IM,NK) == NN ) THEN

IP = IM + 1

XM = PTL(NN,4)-REAL(IM)

XP = 1.0 - XM

TRYm(NK,IM,1) = TRYm(NK,IM,1) + XM * PTL(NN,1)

TRYp(NK,IP,1) = TRYp(NK,IP,1) + XP * PTL(NN,1)

TRYm(NK,IM,2) = TRYm(NK,IM,2) + XM * PTL(NN,2)

TRYp(NK,IP,2) = TRYp(NK,IP,2) + XP * PTL(NN,2)

TRYm(NK,IM,3) = TRYm(NK,IM,3) + XM * PTL(NN,3)

TRYp(NK,IP,3) = TRYp(NK,IP,3) + XP * PTL(NN,3)

TRYm(NK,IM,4) = TRYm(NK,IM,4)

TRYp(NK,IP,4) = TRYp(NK,IP,4)

ELSE

ct = ct + 1

CIN(ct) = nn

ENDIF

ENDDO

ENDDO

ca = ct

ENDDO

!!

!! REDUCTION  LOOP

!!

DO IX = 1, IMAX

DO IC = 1, 4

DO NK = 1, KVEC

MOM(IX,IC)= MOM(IX,IC) + TRYm(NK,IX,IC) + TRYp(NK,IX,IC)

ENDDO

ENDDO

ENDDO

3．速度比較

作業配列法・LISTVEC法・RETRY法のそれぞれについ

て，計算速度を測定した．粒子はシステム内にランダムに

配置している．測定値は100回測定の平均値である．

3.1．LISTVEC指示行法と作業配列法

作業配列法とLISTVEC指示行法の計算に要した時間

（それぞれTW，TLとする）を比べ，格子点数依存を表示し

たものを図7に示す．粒子は，格子当り平均49個分布させ

ている．上から，全計算時間の比と実時間，メインルー

プ，初期化とreductionに要した時間の和，のそれぞれに関

して，LISTVEC指示行法を用いて要した計算時間（赤線）

と作業配列法を用いて要した計算時間（黒線）を表示し

ている．結果，計算に要する時間は，常時，作業配列法

が短い．しかし，格子点の増加に伴い，LISTVEC指示行

法においては衝突発生確率が下がるためと，作業配列法

における初期化とreduction作業コストが高くなる事によ

り，格子点の増加とともに，計算時間比（TL/TW）が小さ

くなる事が見てとれ，約1.5である（最上段パネル）．

ベクトル型計算機によるPICモーメント計算法の速度比較
Performance measurement of PIC moment calculation

Fig. 7.   Executing time for LISTVEC and WORK ARRAY algorithm. Dependence on the total number of grid are plotted. From top to bottom, the

executing time of total calculation, main loop calculation and others of initialization plus reduction. The executing time of "others" has large calculation

cost relative to total cost which is almost 7 %.

図7. LISTVEC指示行法（赤線）と作業配列法（黒線）を用いた計算に要した時間．上から，全計算時間・メインループ，初期化と
reductionの和，に要した時間を格子数に対してプロット．作業配列法においては，格子数が増えると初期化とreduction に要する時間が全体
の7％にも達する．



3.2．RETRY法の計算時間特性

RETRY法プログラムである図６において，変数RVECと

KVECは，互いに独立であり，かつ，使い方は作業配列法

におけるLVECと同じであるが，この変数の大きさは，ベ

クトル長とは関係が無い．よって，RETRY法の計算時間

のRVECとKVECに対する依存性は自明ではない．図8にメ

インループ計算時間（TMR）に関するKVECとRVECに対す

る依存性を示す．ここでは，純粋に，粒子ID番号の格子

点位置への書き込み衝突頻度に関する依存を見るため，

初期化とreductionを除いたメインループに対して計測す

る．格子数は6×105個，粒子は格子当り平均199個分布さ

せている．縦軸の値は，作業配列法の計算時間（TMW）に

対する比（TMR/TMW）をプロットしている．上図は，

RVEC=256と2,048とに固定した時のKVEC依存である．明

らかに，KVECが2の冪上の時が高速に計算されており，

中でもKVEC=32の時が最速である．下図は，KVEC=2, 32

に固定した時のRVEC依存である．RVEC=256の場合の計

算時間が他の場合より遅くなっていることから，RVECが

ベクトル長と関係ない事が分かる．また，計算時間比

（TMR/TMW）の値が1.0～1.1程度であることから，LISTVEC

法より高速に計算可能である．

KVECの値を大きくすれば，作業配列法のLVECの設置

と同じく主メモリを圧迫するため，可能な限り小さい値

が良い．LISTVEC法より使用する主メモリ量を同程度か

少なく，かつ高速計算を実現することを考慮して，今後

はKVEC=2，RVEC=2,048を採用する．

Fig. 8.   Executing time for RETRY algorithm. Dependence on KVEC (top) and RVEC (bottom) parameters are plotted. The speed is fastest at KVEC=32.

RVEC=256 is not good for RETRY algorithm even though the value corresponds to the length of the vector register.

図8. RETRY法におけるパラメータKVECとRVECに対する計算時間依存性．時間は作業配列法の計算時間との比（TMR/TMW）がプロットさ
れている．（上図）KVECに対しては，2の冪上の値をとる時が高速であり，KVEC=32の時が最速であるが，主メモリ使用量を減らすため，
なるべく小さい値を用いる方が良い．（下図）RVECに対しては，最大ベクトル長と同じ長さのRVEC=256の時は，他の場合より遅い事が見
てとれる．

ベクトル型計算機によるPICモーメント計算法の速度比較
Performance measurement of PIC moment calculation

Fig. 9.   Contour plot for the executing time ratio relative to WORK ARRAY algorithm in space of number of particle per cell versus total number of grid.

図9. LISTVEC法とRETRY法の計算時間について，作業配列法との比の値に対して等高線表示している．横軸は総粒子数．縦軸は格子点
当りの平均粒子分布数．

Fig. 10.   Contour plot for used main memory in the same format as Fig. 9.

図10. LISTVEC法とRETRY法の使用メモリ量について，作業配列法との比の値に対して等高線表示している．図9とフォーマットは同じ．

3.3．全計算時間における速度比較

上記3.1節で述べたように，計算速度は格子数や粒子数

に依存する．よって，両変数に対する計算速度比の依存

性を図9に示す．表示している値は，全計算時間の，作業

配列法の計算時間に対する比をプロットしている．左図

のLISTVEC法に関しては，右下の領域，すなわち，粒子

数が少なく格子点数が多くなる程，計算速度比が向上す

る．これは，LISTVEC法のメインループ内で衝突確率が

減少することと，作業配列法において，格子点数が増加

すると初期化とreductionの計算時間が長くなるためであ

る．右図のRETRY法に関しては，粒子数が少ない程，計

算速度が速くなるが，全体的に作業配列法と同じ速度で

計算出来ていることが最大の特徴である．図10に，計算

に使用する主メモリ量に関して，作業配列法との比を図9

と同じ書式で示す．KVEC=2を採用したため，RETRY法に

おける使用主メモリ量はLISTVEC法とほぼ同じである．

4．まとめ

これまで，作業配列法が広くプラズマPICシミュレーシ

ョンに使われてきた．そこでは，計算速度を上げるために，

ベクトル長の256と同じ大きさの作業配列を確保していた

ため，主メモリを圧迫していた．256要素数の配列を確保

することは，その数だけの粒子を格子当りに分布させるこ

とと同値である．通常，格子当り100個程度の粒子を使用

するPICシミュレーションにおいては，本末転倒である．

そこで，主メモリ使用量を抑えたLISTVEC指示行法・

RETRY法が提案されている．これらの主メモリ使用量は，

典型的なパラメータである粒子数が格子当り100個，格子

総数106～107個では，作業配列法の40％以下である（図10）．

今回，これらの手法を用いた場合の計算時間の計測を行っ

た結果，LISTVEC法では計算時間が約1.5倍に，RETRY法

では1.0~1.1倍長くなるだけと，ほぼ作業配列法と変わらな

い計算速度であることが確認された．RETRY法のプログラ

ム作成は，他の２つに比べ煩雑であるが，その作成によっ

て得られる主メモリの節約という効果が大きいため，使用

を強く推薦する．

今回は，2次元計算におけるArea-Weighting手法のモーメ

ント値計算に関して，計算速度を測定した．プラズマPIC

計算では，1次のモーメント値を得る方法に電荷保存法も

存在する．次回は，電荷保存法について検討したい．

